

NAG Library Routine Document

F07BFF (DGBEQU)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F07BFF (DGBEQU) computes diagonal scaling matrices D_R and D_C intended to equilibrate a real m by n band matrix A of band width $(k_l + k_u + 1)$, and reduce its condition number.

2 Specification

```
SUBROUTINE F07BFF (M, N, KL, KU, AB, LDAB, R, C, ROWCND, COLCND, AMAX,      &
                  INFO)
```

```
INTEGER          M, N, KL, KU, LDAB, INFO
REAL (KIND=nag_wp) AB(LDAB,*), R(M), C(N), ROWCND, COLCND, AMAX
```

The routine may be called by its LAPACK name *dgbequ*.

3 Description

F07BFF (DGBEQU) computes the diagonal scaling matrices. The diagonal scaling matrices are chosen to try to make the elements of largest absolute value in each row and column of the matrix B given by

$$B = D_R A D_C$$

have absolute value 1. The diagonal elements of D_R and D_C are restricted to lie in the safe range $(\delta, 1/\delta)$, where δ is the value returned by routine X02AMF. Use of these scaling factors is not guaranteed to reduce the condition number of A but works well in practice.

4 References

None.

5 Arguments

- | | | |
|----|---|--------------|
| 1: | M – INTEGER | <i>Input</i> |
| | <i>On entry:</i> m , the number of rows of the matrix A . | |
| | <i>Constraint:</i> $M \geq 0$. | |
| 2: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> n , the number of columns of the matrix A . | |
| | <i>Constraint:</i> $N \geq 0$. | |
| 3: | KL – INTEGER | <i>Input</i> |
| | <i>On entry:</i> k_l , the number of subdiagonals of the matrix A . | |
| | <i>Constraint:</i> $KL \geq 0$. | |
| 4: | KU – INTEGER | <i>Input</i> |
| | <i>On entry:</i> k_u , the number of superdiagonals of the matrix A . | |
| | <i>Constraint:</i> $KU \geq 0$. | |

- 5: AB(LDAB,*) – REAL (KIND=nag_wp) array Input
Note: the second dimension of the array AB must be at least $\max(1, N)$.
On entry: the m by n band matrix A whose scaling factors are to be computed.
 The matrix is stored in rows 1 to $k_l + k_u + 1$, more precisely, the element A_{ij} must be stored in

$$AB(k_u + 1 + i - j, j) \quad \text{for } \max(1, j - k_u) \leq i \leq \min(m, j + k_l).$$
 See Section 9 in F07BAF (DGBSV) for further details.
- 6: LDAB – INTEGER Input
On entry: the first dimension of the array AB as declared in the (sub)program from which F07BFF (DGBEQU) is called.
Constraint: $LDAB \geq KL + KU + 1$.
- 7: R(M) – REAL (KIND=nag_wp) array Output
On exit: if INFO = 0 or INFO > M, R contains the row scale factors, the diagonal elements of D_R . The elements of R will be positive.
- 8: C(N) – REAL (KIND=nag_wp) array Output
On exit: if INFO = 0, C contains the column scale factors, the diagonal elements of D_C . The elements of C will be positive.
- 9: ROWCND – REAL (KIND=nag_wp) Output
On exit: if INFO = 0 or INFO > M, ROWCND contains the ratio of the smallest value of $R(i)$ to the largest value of $R(i)$. If $ROWCND \geq 0.1$ and AMAX is neither too large nor too small, it is not worth scaling by D_R .
- 10: COLCND – REAL (KIND=nag_wp) Output
On exit: if INFO = 0, COLCND contains the ratio of the smallest value of $C(i)$ to the largest value of $C(i)$.
 If $COLCND \geq 0.1$, it is not worth scaling by D_C .
- 11: AMAX – REAL (KIND=nag_wp) Output
On exit: $\max |a_{ij}|$. If AMAX is very close to overflow or underflow, the matrix A should be scaled.
- 12: INFO – INTEGER Output
On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = $-i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0 and INFO ≤ M

Row $\langle value \rangle$ of A is exactly zero.

INFO > M

Column $\langle value \rangle$ of A is exactly zero.

7 Accuracy

The computed scale factors will be close to the exact scale factors.

8 Parallelism and Performance

F07BFF (DGBEQU) is not threaded in any implementation.

9 Further Comments

The complex analogue of this routine is F07BTF (ZGBEQU).

10 Example

This example equilibrates the band matrix A given by

$$A = \begin{pmatrix} -0.23 & 2.54 & -3.66 \times 10^{-10} & 0 \\ -6.98 \times 10^{10} & 2.46 \times 10^{10} & -2.73 & -2.13 \times 10^{10} \\ 0 & 2.56 & 2.46 \times 10^{-10} & 4.07 \\ 0 & 0 & -4.78 \times 10^{-10} & -3.82 \end{pmatrix}.$$

Details of the scaling factors, and the scaled matrix are output.

10.1 Program Text

```

Program f07bffe

!      F07BFF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
!      Use nag_library, Only: dgbequ, dscal, f06fcf, nag_wp, x02ajf, x02amf,      &
!                               x02bhf, x04cef

!      .. Implicit None Statement ..
!      Implicit None

!      .. Parameters ..
!      Real (Kind=nag_wp), Parameter      :: one = 1.0_nag_wp
!      Real (Kind=nag_wp), Parameter      :: thresh = 0.1_nag_wp
!      Integer, Parameter                  :: nin = 5, nout = 6

!      .. Local Scalars ..
!      Real (Kind=nag_wp)                  :: amax, big, colcnd, rowcnd, small
!      Integer                              :: i, i0, il, ifail, ilen, info, j, k, &
!                                           kl, ku, ldab, n

!      .. Local Arrays ..
!      Real (Kind=nag_wp), Allocatable     :: ab(:,,:), c(:), r(:)

!      .. Intrinsic Procedures ..
!      Intrinsic                          :: max, min, real

!      .. Executable Statements ..
!      Write (nout,*) 'F07BFF Example Program Results'
!      Write (nout,*)
!      Flush (nout)
!      Skip heading in data file
!      Read (nin,*)
!      Read (nin,*) n, kl, ku
!      ldab = kl + ku + 1
!      Allocate (ab(ldab,n),c(n),r(n))

!      Read the band matrix A from data file

!      k = ku + 1
!      Read (nin,*)((ab(k+i-j,j),j=max(i-kl,1),min(i+ku,n)),i=1,n)

!      Print the matrix A

!      ifail: behaviour on error exit

```

```

!           =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call x04cef(n,n,kl,ku,ab,ldab,'Matrix A',ifail)

Write (nout,*)

! Compute row and column scaling factors

! The NAG name equivalent of dgbequ is f07bff
Call dgbequ(n,n,kl,ku,ab,ldab,r,c,rowcnd,colcnd,amax,info)

If (info>0) Then
  If (info<=n) Then
    Write (nout,99999) 'Row ', info, ' of A is exactly zero'
  Else
    Write (nout,99999) 'Column ', info - n, ' of A is exactly zero'
  End If
Else

! Print ROWCND, COLCND, AMAX and the scale factors

Write (nout,99998) 'ROWCND =', rowcnd, ', COLCND =', colcnd,      &
', AMAX =', amax
Write (nout,*)
Write (nout,*) 'Row scale factors'
Write (nout,99997) r(1:n)
Write (nout,*)
Write (nout,*) 'Column scale factors'
Write (nout,99997) c(1:n)
Write (nout,*)
Flush (nout)

! Compute values close to underflow and overflow

small = x02amf()/(x02ajf()*real(x02bhf(),kind=nag_wp))
big = one/small
If ((rowcnd>=thresh) .And. (amax>=small) .And. (amax<=big)) Then
  If (colcnd<thresh) Then
! Just column scale A
! The NAG name equivalent of dscal is f06edf
  Do j = 1, n
    il = 1 + max(1,j-ku) - (j-ku)
    ilen = min(n,j+kl) - max(1,j-ku) + 1
    Call dscal(ilen,c(j),ab(il,j),1)
  End Do

  End If
  Else If (colcnd>=thresh) Then

! Just row scale A
  Do j = 1, n
    i0 = max(1,j-ku)
    il = 1 + i0 - (j-ku)
    ilen = min(n,j+kl) - i0 + 1
    Call f06fcf(ilen,r(i0),1,ab(il,j),1)
  End Do

  Else

! Row and column scale A
  Do j = 1, n
    i0 = max(1,j-ku)
    il = 1 + i0 - (j-ku)
    ilen = min(n,j+kl) - i0 + 1
    Call dscal(ilen,c(j),ab(il,j),1)
    Call f06fcf(ilen,r(i0),1,ab(il,j),1)
  End Do

  End If

! Print the scaled matrix

```

```

      ifail = 0
      Call x04cef(n,n,kl,ku,ab,ldab,'Scaled matrix',ifail)

      End If

99999 Format (1X,A,I4,A)
99998 Format (1X,3(A,1P,E8.1))
99997 Format ((1X,1P,7E11.2))
      End Program f07bffe

```

10.2 Program Data

F07BFF Example Program Data

```

 4  1  2                               :Values of N, KL and KU
-2.30D-01  2.54D+00 -3.66D-10
-6.98D+10  2.46D+10 -2.73D+00 -2.13D+10
           2.56D+00  2.46D-10  4.07D+00
           -4.78D-10 -3.82D+00       :End of matrix A

```

10.3 Program Results

F07BFF Example Program Results

Matrix A

	1	2	3	4
1	-2.3000E-01	2.5400E+00	-3.6600E-10	
2	-6.9800E+10	2.4600E+10	-2.7300E+00	-2.1300E+10
3		2.5600E+00	2.4600E-10	4.0700E+00
4			-4.7800E-10	-3.8200E+00

ROWCND = 3.6E-11, COLCND = 1.4E-10, AMAX = 7.0E+10

Row scale factors

3.94E-01	1.43E-11	2.46E-01	2.62E-01
----------	----------	----------	----------

Column scale factors

1.00E+00	1.00E+00	6.94E+09	1.00E+00
----------	----------	----------	----------

Scaled matrix

	1	2	3	4
1	-0.0906	1.0000	-1.0000	
2	-1.0000	0.3524	-0.2714	-0.3052
3		0.6290	0.4195	1.0000
4			-0.8684	-1.0000
